

TUTORIALREIHE WPF LERNEN

WPF verständlich und anhand von Praxisbeispielen erklärt

Sascha-Heinz Patschka

Inhaltsverzeichnis (inkrementell)

1. Vorstellung

1.1. Einleitung

1.2. Aufbau dieser Tutorialreihe

2. Grundlagen der WPF

2.1. Einführung in die WPF

2.1.1. Die wichtigsten Controls und deren Verwendung

2.1.1.1. Die wichtigsten Controls und ihr Verhalten

2.1.1.2. Styles, Templates und Trigger

2.1.1.3. Controls eine neue Optik verpassen

2.1.2. XAML Namespaces

2.1.2.1. Kurze Theorie

2.1.2.2. Erstellung eigener Namespaces

2.1.3. Ressourcen

2.1.3.1. Was sind Ressourcen, was bringen sie mir

2.1.3.2. Unterschied StaticResource und DynamicResource

2.1.4. Binding und das Bindingsystem

2.1.4.1. Was ist DataBinding? Das Konzept dahinter + Videocast

2.1.4.2. Binding anhand einfacher Beispiele und Klassen

2.1.4.3. DesignTime-Support für Binding

2.1.4.4. Binding über Converter

2.1.4.5. Binding über DataTemplates

2.1.4.6. Binding an Collections. Warum ICollectionViewSource? Collections Filter, Sortieren, Gruppieren ohne viel Aufwand

2.1.4.7. Validierung von Benutzereingaben

2.1.4.8. Rücksicht nehmen auf die aktuelle Culture

2.1.5. DependencyProperties

2.1.5.1. Was sind Dependency Properties und wie unterscheiden sie sich von normalen Properties

- 2.1.5.2. *Eigene Dependency Properties implementieren*
- 2.1.6. *Markuperweiterungen*
 - 2.1.6.1. *Kurze Theorie*
 - 2.1.6.2. *Beispiele Anhand von Ressourcen und Styles*
- 2.1.7. *Attached Properties*
 - 2.1.7.1. *Kurze Theorie (wozu Attached Properties)*
 - 2.1.7.2. *Beispiele Anhand vom Grid und dem DockPanel*
 - 2.1.7.3. *Eigene Attached Properties erstellen ;-)*
- 2.1.8. *Attached Events*
 - 2.1.8.1. *Wir gehen gleich in die Praxis, gibt nicht viel zu sagen*
- 2.1.9. *Inputs und Command*
 - 2.1.9.1. *Die Input-API*
 - 2.1.9.2. *Tastatur und Mausklassen*
 - 2.1.9.3. *Eventrouting (Direct, Bubbling, Tunneling)*
 - 2.1.9.4. *Keyboard, Mouse und Textinput*
 - 2.1.9.5. *Touch und Multitouch (wird ja immer wichtiger)*
 - 2.1.9.6. *Focus (Der Unterschied zwischen Keyboardfocus und Logicalfocus)*
 - 2.1.9.7. *Commands (Integrierte und eigene)*
 - 2.1.9.8. *RelayCommand Klasse*
 - 2.1.9.9. *CommandBinding und Command-Parameter*
- ~~3. Eine Telefonbuch Applikation unter WPF (ohne Binding)~~
 - ~~3.1. Hauptfenster erstellen und Funktionen festlegen~~
 - ~~3.2. Ein Primitives Telefonbuch rein mit CodeBehind ala WinForms~~
 - ~~3.3. Fazit~~
- 4. *Eine Telefonbuch Applikation unter WPF (mit Binding der CodeBehind)*
 - 4.1. *Das Model erstellen / was soll die App können*
 - 4.2. *Die erste View erstellen und Binden*
 - 4.3. *Die erste View besser strukturieren*
 - 4.4. *Filtermöglichkeit einbinden (oder doch mehr?)*
 - 4.5. *Einträge bearbeiten und speichern*
- 5. *Das MVVM Pattern*
 - 5.1. *Was ist das MVVM Pattern?*
 - 5.2. *Wann MVVM und wann nicht?*
 - 5.3. *Welchen Mehrwert kann ich aus dem Pattern gewinnen?*
 - 5.4. *MVVM und CodeBehind – verboten?*
 - 5.5. *Model – View – ViewModel – Wars das?*
 - 5.6. *Erstellen einer korrekten MVVM Projektmappe in Visual Studio*
- 6. *Unser Telefonbuch in MVVM*
 - 6.1. *Projekt anlegen und Struktur besprechen*

- 6.2. Das Model erstellen**
- 6.3. ViewModel - Der Core – Was benötigen wir alles ehe wir anfangen mit unserem Programm**
- 6.4. Wie MessageBox, Dialoge, Mousecursor oder TaskbarInfo steuern wenn ich die View nicht kenne**
- 6.5. Jetzt anfangen? Ne? Warum?**
- 6.6. Das MainViewModel erstellen**
- 6.7.**
- 6.8. Fazit zum MVVM Pattern**
- 7. Lokalisierung und Globalisierung**
 - 7.1. Lokalisierung nur mit Boardmitteln**
 - 7.2. Lokalisierung mit schwung (unter Zuhilfenahme von zwei NuGet-Paketen)**
 - 7.3. Globalisierung (Datum, Währung, usw.)**
 - 7.4. Lokalisieren von Werten aus Fremssystemen (DB, XML usw.)**
 - 7.5. Gute Hilfsprogramme und Helferlein**
- 8. UnitTest und IntegrationTests**
 - 8.1. Wozu UnitTests?**
 - 8.2. Wie schreibe ich Tests (Grundlagen)**
 - 8.3. Testen unseres ViewModels möglich?**
 - 8.4. ...**
 - 8.5. ...**
 - 8.6. Fazit**
- 9. Businesslogic**
 - 9.1. Warum besser nicht im ViewModel**
 - 9.2. Vor und Nachteile einer Businesslogik**
 - 9.3. Neubau unseres Telefonbuchs mit eigener Businesslogik**
 - 9.4. Fazit**
- 10. Repository (DataAccessLayer)**
 - 10.1. Noch einen Schritt weiter? Wozu?**
 - 10.2. Besprechen und Aufsetzen eines Repository`s**
 - 10.3. Wir bauen unser Telefonbuch abermals neu ;-(**
 - 10.4. Fazit**

4.0

Eine WPF Telefonbuch App

Jetzt kommen wir zum praktischen Teil. Wir erstellen nun eine kleine App unter Verwendung des bisher gelerntem. Wir werden noch nicht mit MVVM oder ähnlichem Pattern in Berührung kommen, das ist aber gewollt so. Es ist besser wenn wir uns noch nicht mit komplizierten Pattern Beschäftigen und erstmal lernen wie wir nun z.b. von Binding profitieren können.

Wir haben ja gelernt dass wir wunderbar auf Klassen Binden können und hier ein wirklich Leistungsstarkes System an die Hand bekommen um schnell und unkompliziert Daten mit Controls zu verbinden. Und genau das versuchen wir nun in der Praxis.


Es wird eine Telefonbuch App.

Da ich ungerne Beispiele mache welche nicht wirklich funktionsfähig sind soll die App - auch wenn sie nur diesem Tutorial dienen soll - Daten speichern können. Also soll auch die persistierung funktionieren, damit wir hier flexibel sind werde ich Klassen haben welche schlicht Serialisiert werden. Wir könnten in diese „Modelklassen“ natürlich theoretisch [INotifyPropertyChanged](#) implementieren um diese direkt als eine Art „ViewModel“ zu missbrauchen. Ich bin aber kein Freund davon. Warum werden wir im Laufe dieses Kapitel noch näher erfahren. Nun gut, fangen wir mit dem Model eines Telefonbuchs an.

4.1

Das Model erstellen / was soll die App können

Wir erstellen eine neue WPF Solution in welcher wir erstmal einen Order „Model“ anlegen. In diesem landen unsere Modelklassen.



Wir legen eine neue Klasse „[Contact](#)“ an und fügen folgende typische Properties ein.

1

```
Namespace Model
```

```
Public Class Contact
```

```
Public Property FirstName As String  
Public Property LastName As String  
Public Property ImagePath As String  
Public Property Birthday As Date?  
Public Property Note As String
```

```
End Class
```

```
End Namespace
```

Ich denke diese erklären sich von selbst und sind typisch für einen Kontakt eines Telefonbuchs.

Nun hätten wir gerne noch eine Eigenschaft um die Telefonnummern oder Mailadressen zu hinterlegen. Das können ja mehrere je Kontakt sein. Also ein Kontakt kann beispielsweise zwei Mailadressen und eine Telefonnummer haben. Aber vielleicht auch eine Website und einen SocialMedia Account wie Facebook oder Instagram.

Wir benötigen hierfür eine Auflistung. Also legen wir uns eine Eigenschaft vom Typ [List\(Of T\)](#) an. Dafür benötigen wir aber erstmal eine neue Klasse. Nennen wir diese [ContactData](#).

```
Namespace Model
```

```
Public Class ContactData
```

```
Public Property Data As String  
Public Property DataType As EnumContactDataType
```

```
End Class
```

```
Public Enum EnumContactDataType
```

```
Phonenumber = 0  
Mail = 1  
SocialMedia = 2  
Fax = 3  
Website = 4
```

```
End Enum
```

```
End Namespace
```

Diese Klasse hat zwei Eigenschaften. Zum einen die Daten vom Typ String. Denn es sollen sowohl Telefonnummern aber auch z.B. Mailadressen gespeichert werden können. Da ist String der einzige Datentyp der hierfür geeignet ist. Um aber später unterscheiden zu können um welche Art von Daten es sich hierbei handelt habe ich einen Enumerator erstellt und eine Eigenschaft welche den Wert dieses Enumerators hält. So können wir später die Daten richtig zuordnen.

Nun können wir diese als Auflistung in unsere Contact Klasse einbinden.

2

Namespace Model

```
Public Class Contact
```

```
Public Property FirstName As String
```

```
Public Property LastName As String
```

```
...
```

```
Public Property ContactData As List(Of ContactData)
```

```
End Class
```

```
End Namespace
```

Tja, ich würde sagen das reicht uns fürs erste erstmal als Model für unser Telefonbuch. Ich weiß das Kapitel war etwas kurz, aber keine Sorge, beim nächsten geht's ran an den Speck.

Hier geht's zum Video: <https://youtu.be/2eWaAs1w9ok>

Hier zum Thread im VB-Paradise-Forum: <https://www.vb-paradise.de/index.php/Thread/124617-Tutorialreihe-WPF-lernen>

Hier zum SupportThread im Forum: <https://www.vb-paradise.de/index.php/Thread/124641-Support-Anregungen-W%C3%BCnsche-zur-Tutorialreihe-WPF-lernen/>

4.2

Die erste View erstellen

In diesem Kapitel werden wir unsere erste View mit einer Klasse verbinden. Wie im letzten Kapitel bereits erwähnt werde ich hier nicht direkt mit den Modelklassen arbeiten, ihr werdet im Verlauf der nächsten Kapitel – aber auch in diesem – sehen warum dies von Vorteil sein kann.

Als legen wir uns einen Order für unsere „[ViewModel](#)“-Klassen an. Ich nenne diese ViewModel-Klassen auch wenn es hier nicht um MVVM geht. Das eine hat mit dem anderen ja nicht viel zu tun. Jede Klasse welche mit einem View verbunden wird kann als ViewModel-Klasse bezeichnet werden.

Als erste Klasse lege ich eine Basisklasse an damit wir nicht für jede einzelne Klasse [INotifyPropertyChanged](#) implementieren müssen sondern eine Basisklasse haben von welcher wir immer wieder erben können. Das spart uns viel Tipparbeit. Die ViewModel-Klasse packe ich alle in einen separaten Namespace „[ViewModel](#)“.

3

```

Namespace ViewModel
    Public MustInherit Class ViewModelBase
        Implements INotifyPropertyChanged

        Protected Overridable Sub RaisePropertyChanged(<CallerMemberName> Optional
prop As String = "")
            RaiseEvent PropertyChanged(Me, New PropertyChangedEventArgs(prop))
        End Sub

        Public Event PropertyChanged As PropertyChangedEventHandler Implements
INotifyPropertyChanged.PropertyChanged
    End Class
End Namespace

```

Nun können wir bereits beginnen die erste ViewModel-Klasse anzulegen. Die `ContactViewModel`-Klasse. Diese Klasse hat die Aufgabe die Eigenschaften der Model-Klasse „Contact“ für die View aufzubereiten. Wie schon erwähnt machen es viele so das sie `INotifyPropertyChanged` in der Modelklasse implementieren um sich so diese Zwischenklasse zu „sparen“. Das ist natürlich ein Weg den man gehen kann um dem *dont repeat yourself* Grundsatz zu folgen, ich habe aber eben die Erfahrung gemacht das sich diese zusätzlicher Arbeit durchaus auszahlt und sogar eher komplexität herausnimmt – auch wenn dies auf den ersten Blick nicht so aussieht - speziell auf Hinblick auf späteres MVVM, aber das ist ein späteres Thema, ihr werdet auch in diesem Kapitel bereits sehen das dies von Vorteil sein kann.

Erstellen wir erstmal eine Klasse, ich erkläre hierzu gleich ein paar Dinge:

```

Namespace ViewModel
    Public Class ContactViewModel
        Inherits ViewModelBase

        Private ReadOnly _modelContact As Model.Contact

        Public Sub New()
            _modelContact = New Model.Contact
            ContactData = New ObservableCollection(Of ContactDataViewModel)
        End Sub

        Friend Sub New(model As Model.Contact)
            _modelContact = model
            ContactData = New ObservableCollection(Of ContactDataViewModel)
            model.ContactData.ForEach(Sub(x) ContactData.Add(New ContactDataViewModel(x)))
        End Sub

        Public Property FullName() As String
            Get
                Return $"{_modelContact.FirstName} {_modelContact.LastName}"
            End Get
            Set(ByVal value As String)
                Dim fullname = value.Split(" ")
                _modelContact.FirstName = fullname.First
                If fullname.Length > 1 Then _modelContact.LastName = fullname.Last
                RaisePropertyChanged()
            End Set
        End Property

        Public Property ImagePath As String
            Get
                Return _modelContact.ImagePath
            End Get
            Set(value As String)
                _modelContact.ImagePath = value
                RaisePropertyChanged()
            End Set
        End Property
        Public Property Birthday As Date?
            Get
                Return _modelContact.Birthday
            End Get
            Set(value As Date?)
                _modelContact.Birthday = value
                RaisePropertyChanged()
            End Set
        End Property
        Public Property Note As String
            Get
                Return _modelContact.Note
            End Get
            Set(value As String)
                _modelContact.Note = value
                RaisePropertyChanged()
            End Set
        End Property

        Private _contactData As ObservableCollection(Of ContactDataViewModel)
        Public Property ContactData() As ObservableCollection(Of ContactDataViewModel)
            Get
                Return _contactData
            End Get
            Set(ByVal value As ObservableCollection(Of ContactDataViewModel))
                _contactData = value
                RaisePropertyChanged()
            End Set
        End Property
    End Class
End Namespace

```

5

OK, das sieht erstmal verwirrend aus. Gebe ich zu.

Fangen wir oben an. Wir haben eine Private Schreibgeschützte Variable vom Typ `Model.Contact`.

Diese Variable dient dazu unser „Modelobjekt“ zu halten. Wie an den beiden Konstruktoren zu sehen, wird entweder eine neue Instanz eines `Contact` erstellt falls beim Instanzieren unseres ViewModels keine Instanz hereingereicht wird, oder es wird die Instanz an die Variable übergeben. Wir haben somit immer ein Model-Object in dieser Variable.

In späterer folge haben wir dann eine Eigenschaft `FullName`. Hier ist schon mal ein Fall für den eine ViewModel Klasse eben praktisch ist. In unserem Model ist definiert das wir `FirstName` und `LastName` getrennt speichern werden. Wie Ihr es vom Handy evtl. kennt kann man einen neuen Kontakt aber meist vereinfacht eingeben indem man einfach den Vor und den Zunamen in einer Zeile eingibt. Tja, unser Telefonbuch soll dies auch können, gespeichert soll aber trotzdem getrennt werden. Also wenn der User „Sascha Patschka“ in das Feld eingibt soll unsere App hier „Sascha“ als Vornamen und „Patschka“ als Nachnamen speichern. Das ist bereits der Vorteil der Trennung zwischen einem *ViewModel* und einem *Model*. Wir können frei bestimmen und sind flexibel wie die View aussehen soll bzw. wie Daten sowohl angezeigt als auch eingegeben werden können.

Im Getter der Eigenschaft setzen wir die Model-Eigenschaften `FirstName` und `LastName` zusammen und geben diese mit einem Leerzeichen getrennt aus.

Im Setter lesen wir den String aus und versuchen diesen zu Splitten und anschliessend getrennt an das Modelobjekt zu übergeben . Die Art wie dies geschied ist im Moment nur sehr rudimentär implementiert und hat noch so einige Fehler, reicht uns aber fürs erste um einfach hier nur eine Idee zu bekommen was wir hier vorhaben.

Nun sehen wir uns die nächsten Eigenschaften an. Diese reichen im Grunde einfach nur durch.

```
Public Property ImagePath As String
    Get
        Return _modelContact.ImagePath
    End Get
    Set(value As String)
        _modelContact.ImagePath = value
        RaisePropertyChanged()
    End Set
End Property
```

Im Getter wir der `ImagePath` des ModelObjekts einfach durchgereicht und im Setter wird `ImagePath` des Modelobjekts gesetzt. Man kann dies als Wrapper bezeichnen.

6

Im Grunde geht es einfach darum eine Eigenschaft zu haben welche nach dem setzen des Wertes das Event `PropertyChanged` wirft und das macht unsere Eigenschaft im Setter mit der Methode `RaisePropertyChanged()`.

Kommen wir zur letzten Eigenschaft „`ContactData`“ vom Typ `ObservableCollection(Of ContactDataViewModel)`.

Diese in Verbindung mit der dritten Zeile des Konstruktors ist etwas speziell und kann vielleicht verwirrend sein. Aber keine Sorge, wir gehen das langsam durch.

Erstmal benötigen wir den Typ `ContactDataViewModel`.

```
Namespace ViewModel
    Public Class ContactDataViewModel
        Inherits ViewModelBase

        Private ReadOnly _modelContactData As ContactData

        Public Sub New()
            _modelContactData = New ContactData()
        End Sub

        Friend Sub New(model As Model.ContactData)
            _modelContactData = model
        End Sub

        Public Property Data As String
            Get
                Return _modelContactData.Data
            End Get
            Set(value As String)
                _modelContactData.Data = value
                RaisePropertyChanged()
            End Set
        End Property
        Public Property DataType As EnumContactDataType
            Get
                Return _modelContactData.DataType
            End Get
            Set(value As EnumContactDataType)
                _modelContactData.DataType = value
                RaisePropertyChanged()
            End Set
        End Property

    End Class
End Namespace
```

Diese Klasse sollte soweit verständlich sein, auch diese dient als „Wrapper“ für ein Model-Objekt.

Zurück zur `ContactViewModel`-Klasse...

Wir bekommen ja unser Model-Objekt in die private Variable. Um nun die Auflistung an `ContactData` des Model-Objekts in die Eigenschaft `ContactData` des ViewModels zu bekommen können wir diesmal nicht einfach „durchreichen“ da die Typen sich unterscheiden. Wir müssen

7

die Typen also konvertieren. Das machen wir am einfachsten indem wir in einer Schleife die Auflistung durchgehen.

Da wir in der Klasse [ContactDataViewModel](#) einen Konstruktor haben welchem wir ein Model-Objekt übergeben können ist dies auch einfach. Wir erstellen schlicht für jedes Objekt in der Auflistung des Model-Objekts eine neue Instanz von [ContactDataViewModel](#) und verwenden den Konstruktor mit der Überladung welcher wir ein Modelobjekt mitgeben können was in der dritten Zeile des Konstruktors von [ContactViewModel](#) passiert.

Schon haben wir dafür gesorgt dass die Auflistung mit Daten gefüllt ist.

Nun muss nur noch schnell ein View her um das mal zu testen.

Wir öffnen die CodeBehind des [MainWindow](#) und erstellen erstmal eine Eigenschaft welche eine Hand voll Testdaten halten soll. Eine [ObservableCollection\(Of ContactViewModel\)](#)

```
Class MainWindow
```

```
Private _allContacts As ObservableCollection(Of ContactViewModel)
Public Property AllContacts() As ObservableCollection(Of ContactViewModel)
    Get
        Return _allContacts
    End Get
    Set(ByVal value As ObservableCollection(Of ContactViewModel))
        _allContacts = value
    End Set
End Property
```

```
End Class
```

OK, aber wir müssen die WPF für Binding wieder Benachrichtigen. Die Basisklasse für unsere ViewModels können wir bei einem Window aber nicht erben, wir müssen hier also [INotifyPropertyChanged](#) implementieren.

```
Class MainWindow
```

```
Implements INotifyPropertyChanged
```

```
Private _allContacts As ObservableCollection(Of ContactViewModel)
Public Property AllContacts() As ObservableCollection(Of ContactViewModel)
    Get
        Return _allContacts
    End Get
    Set(ByVal value As ObservableCollection(Of ContactViewModel))
        _allContacts = value
        RaiseEvent PropertyChanged(Me, New PropertyChangedEventArgs(NameOf(AllContacts)))
    End Set
End Property
```

```
Public Event PropertyChanged As PropertyChangedEventHandler Implements
INotifyPropertyChanged.PropertyChanged
End Class
```

Nun Abonnieren wir das [Loaded](#) Event des Window um zum einen ein paar Testdaten zu erstellen und zum anderen die View an die CodeBehind zu Binden.

8

```

Class MainWindow
    Implements INotifyPropertyChanged

    Private Sub MainWindow_Loaded(sender As Object, e As RoutedEventArgs) Handles Me.Loaded
        AllContacts = New ObservableCollection(Of ContactViewModel)

        For i As Integer = 0 To 4
            AllContacts.Add(New ContactViewModel(New Model.Contact() With { .FirstName = "Sascha" & i,
                .LastName = "Patschka",
                .Birthday = New Date(1983, 9, 12), .Note = "Eine
Notiz", .ContactData = New List(Of Model.ContactData) From {
                    New Model.ContactData() With { .Data = "+43 664
1234567", .DataType = Model.EnumContactDataType.Ponenumber}}}))
            Next

        Me.DataContext = Me
    End Sub

    Private _allContacts As ObservableCollection(Of ContactViewModel)
    Public Property AllContacts() As ObservableCollection(Of ContactViewModel)
        Get
            Return _allContacts
        End Get
        Set(ByVal value As ObservableCollection(Of ContactViewModel))
            _allContacts = value
            RaiseEvent PropertyChanged(Me, New PropertyChangedEventArgs(NumberOf(AllContacts)))
        End Set
    End Property

    Public Event PropertyChanged As PropertyChangedEventHandler Implements
INotifyPropertyChanged.PropertyChanged
End Class

```

Wir füllen also `AllContacts` mit Testdaten in einer Schleife. Nun können wir versuchen diese Testdaten mal in einem View anzuzeigen.

```

<Window x:Class="MainWindow"
  xmlns="http://schemas.microsoft.com/winfx/2006/xaml/presentation"
  xmlns:x="http://schemas.microsoft.com/winfx/2006/xaml"
  xmlns:d="http://schemas.microsoft.com/expression/blend/2008"
  xmlns:mc="http://schemas.openxmlformats.org/markup-compatibility/2006"
  xmlns:local="clr-namespace:_4_2_PhoneBook"
  mc:Ignorable="d"
  Title="MainWindow" Height="450" Width="800">
  <Grid>
    <Grid.RowDefinitions>
      <RowDefinition Height="47*"/>
      <RowDefinition Height="372*"/>
    </Grid.RowDefinitions>

    <ItemsControl ItemsSource="{Binding AllContacts}" Grid.Row="1">
      <ItemsControl.ItemTemplate>
        <DataTemplate>


          <TextBlock Text="{Binding FullName}"/>

        </DataTemplate>
      </ItemsControl.ItemTemplate>
    </ItemsControl>

  </Grid>
</Window>

```

Na das sieht doch gut aus. Nicht von der Optik her, sondern das wir korrekt gebunden haben und unser Code bis hierher erstmal funktioniert.



10

Hier geht's zum Video: <https://youtu.be/t-0bwri7wVk>

Hier zum Thread im VB-Paradise-Forum: <https://www.vb-paradise.de/index.php/Thread/124617-Tutorialreihe-WPF-lernen>

Hier zum SupportThread im Forum: <https://www.vb-paradise.de/index.php/Thread/124641-Support-Anregungen-W%C3%BCnsche-zur-Tutorialreihe-WPF-lernen/>

4.3

Die erste View besser strukturieren

Nun werden wir ein wenig Ordnung in den XAML reinbringen. Ich habe abseites des Tutorials ein wenig die Optik der kleinen App verbessert, nichts besonderes aber ein wenig XAML ist entstanden. Gewisse Dinge wurden in Ressourcen ausgelagert und eingebunden aber auch DataTemplates wurden erstellt.

Diesen XAML Code werden wir besser Strukturieren indem wir UserControls erstellen um den Code auszulagern. Ziel ist es in der MailWindow etwas Ordnung zu schaffen.

Da dies nun nicht wirklich Codearbeit ist oder mit Binding sehr viel zu tun hat ist dies hier in Textform etwas schwer zu zeigen weshalb ich mich für dieses Kapitel auf das Video beschränken werde. Ich hoffe ich könnt mir das verzeihen.

Hier geht's zum Video: <https://youtu.be/OdRGPPofn64>

Hier zum Thread im VB-Paradise-Forum: <https://www.vb-paradise.de/index.php/Thread/124617-Tutorialreihe-WPF-lernen>

Hier zum SupportThread im Forum: <https://www.vb-paradise.de/index.php/Thread/124641-Support-Anregungen-W%C3%BCnsche-zur-Tutorialreihe-WPF-lernen/>

4.4

Filtermöglichkeit einbinden (oder doch mehr?)

Wir werden nun eine Filtermöglichkeit für die Kontakte einbinden. Der Benutzer soll ja schließlich nach einem Kontakt suchen können. Ideal wäre hier wenn der Benutzer nach dem Nachnamen, dem Vornamen oder den Informationen in den Notizen suchen könnte.

Genau das implementieren wir nun. Wir haben ja bereits in Kapitel 4.1.4.6 gelernt das uns hier die [ICollectionView](#) zur Verfügung steht um zu Filtern und zu Sortieren.

Wir werden also unsere [MainWindow](#)-CodeBehind anpassen und so umbauen das wir eine [ICollectionView](#) haben auf welche wir Binden können.

Als erstes erstellen wir uns diese und geben unsere bereits vorhandene [ObservableCollection](#) als Datenquelle an.

```
AllContactsView = CollectionViewSource.GetDefaultView(AllContacts)
AllContactsView.Filter = AddressOf Contacts_Filter
```

Wir haben nun auch gleich eine Filtermethode angegeben wie in Kapitel 4.1.4.6 ja bereits gelernt.

In dieser Methode welche ein Boolean als Rückgabewert besitzt, soll nun für jedes Objekt [True](#) zurückgeben in welchem der Name oder die Notiz den eingegebenen Filtertext enthält. Aber hierfür benötigt es noch ein Property für den Filterstring. An diese Eigenschaft binden wir in unserem UserControl [uclMainHeader](#) die dafür vorgesehene Textbox. Wir dürfen aber nicht vergessen im Setter dann die [CollectionView](#) zu aktualisieren.

```
Private _filterString As String = ""
Public Property FilterString() As String
    Get
        Return _filterString
    End Get
    Set(ByVal value As String)
        _filterString = value
        AllContactsView.Refresh()
        RaiseEvent PropertyChanged(Me, New
PropertyChangedEventArgs(NameOf(FilterString)))
    End Set
End Property
```

```
TextBox x:Name="txtSearch" Text="{Binding
FilterString,UpdateSourceTrigger=PropertyChanged}"
        FontSize="{StaticResource DefaultFontSize}"
VerticalContentAlignment="Center" BorderThickness="0"/>
```

Wir ihr sehen könnt habe ich zusätzlich beim Binding den [UpdateSourceTrigger](#) auf

12

`PropertyChanged` gesetzt damit im Code bei jedem Tastenanschlag der Setter durchlaufen wird und wir darauf reagieren können. So ist die Suche für den User weitaus angenehmer.


Nun fehlt uns nur noch die Methode für den eigentlichen Filter zu erstellen.

```
Private Function Contacts_Filter(obj As Object) As Boolean
    Dim c As ContactViewModel = CType(obj, ContactViewModel)
    Return c.FullName.ToLower.Contains(FilterString.ToLower()) Or
    c.Note.ToLower.Contains(FilterString.ToLower())
End Function
```

Nichts wildes, aber effektiv. Jetzt noch das Binding des `ItemsControl` anpassen da wir nun nicht mehr an die `ObservableCollection` sondern an unsere neue `CollectionView` binden möchten.

```
<ItemsControl ItemsSource="{Binding AllContactsView}" Grid.Row="1">
```

Starten wir nun die App können wir bereits Filtern wie es uns gefällt.



Was wir auch noch möchten ist, mit dem X das Suchfeld zu leeren. OK, wir müssen also im Grunde nur die Eigenschaft `FilterString` auf einen Leerstring setzen. Gut. Reagieren wir mal auf einen Klick auf das X.

In unserem `UserControl` haben wir ein `ContentControl`. Wenn wir dieses Klicken soll der Text in der `Textbox` geleert werden. Wer schon ein paar Dinge mit der WPF gemacht hat wird bereits wissen das wenn wir nun auf das `ContentControl` welches als Content einen Path besitzt einen Event verarbeiten wollen ist das unschön. Warum? Der User müsste wirklich genau auf das X klicken. Weil nur wenn die Maus genau über einem Path ist – also genau nur im schwarzen Teil – wird das `Event` verarbeitet. Das ist von der WPF aber so gewollt und ist bei einige Szenarien auch recht wichtig. In unserem Fall möchten wir das in diesem Beispiel aber nicht weshalb wir uns einen `Border` um das `ContentControl` machen und dessen `Background` explizit auf `Transparent` setzen. Tun wir das nicht ist der `Border` erst gar nicht klickbar. In dem `Border` können wir nun das `MouseDown` Event abonnieren und mit `F12` in die CodeBehind des `UserControls` springen.

Unser `DockPanel` sieht nun wie folgt aus:

13


```

<DockPanel LastChildFill="True">
  <ContentControl Margin="5" DockPanel.Dock="Left"
    Content="{StaticResource IconMagnify}"/>
  <Border DockPanel.Dock="Right" MouseLeftButtonDown="Cross_MouseLeftButtonDown"
    Background="Transparent">
    <ContentControl Margin="5" Content="{StaticResource IconCross}"/>
  </Border>
  <TextBox x:Name="txtSearch"
    Text="{Binding FilterString, UpdateSourceTrigger=PropertyChanged}"
    FontSize="{StaticResource DefaultFontSize}" VerticalContentAlignment="Center"
    BorderThickness="0"/>
</DockPanel>

```

In unserer *CodeBehind* haben wir nun folgende Prozedur:

```

Private Sub Cross_MouseLeftButtonDown(sender As Object, e As MouseButtonEventArgs)

End Sub

```

So. Und was jetzt? Wir sind ja in einem *UserControl*. Nicht in dem *MainWindow*. Haben wir uns nun ein Bein gestellt indem wir den XAML ausgelagert haben? Wir können nun die Eigenschaft *FilterString* in der *CodeBehind* der *MainWindow* ja gar nicht setzen? OK, dann geben wir der *TextBox* einen Namen und setzen die *Text*-Eigenschaft dieser *TextBox* einfach auf einen Nullstring. **Halt, Stopp.**

Nein, wir wollen mit Binding arbeiten. Also fangen wir uns das erst gar nicht an. Wir wissen dass dieses *UserControl* auf jeden Fall als Datenkontext eine *MainWindow*-Instanz besitzt. Also werden wir uns diese Instanz holen. Und schon haben wir Zugriff.

```

Dim mw As MainWindow = CType(Me.DataContext, MainWindow)
mw.FilterString = ""

```

Nun funktioniert bereits das Filtern und das leeren des Filters. Super.

Aber wäre es nicht toll nun noch eine Gruppierung zu haben. Bei Handys ist es ja so dass man immer beim Scrollen den Anfangsbuchstaben sieht bei welchem man gerade ist. Also wenn man gerade bei allen Namen welche mit *B* beginnen vorbei ist das dann eine Art Überschrift mit einem *C* kommt. Moment, dafür bietet sich ja eine Gruppierung in Verbindung mit einem *GroupStyle* für das *ItemsControl* an. Gut, legen wir uns einen *Style* an:

```

<ItemsControl ItemsSource="{Binding AllContacts}" Grid.Row="1">
  <ItemsControl.ItemTemplate>
    <DataTemplate>
      <local:uclContactListItemTemplate ContactClicked="ContactItem_Click"/>
    </DataTemplate>
  </ItemsControl.ItemTemplate>
  <ItemsControl.GroupStyle>
    <GroupStyle>
      <GroupStyle.ContainerStyle>
        <Style TargetType="GroupItem">
          <Setter Property="Template">
            <Setter.Value>
              <ControlTemplate TargetType="GroupItem">
                <GroupBox Header="{Binding Name}" >
                  <ItemsPresenter />
                </GroupBox>
              </ControlTemplate>
            </Setter.Value>
          </Setter>
        </Style>
      </GroupStyle.ContainerStyle>
    </GroupStyle>
  </ItemsControl.GroupStyle>
</ItemsControl>

```

Jetzt nur noch wie wir es gelernt haben eine [GroupDescription](#) mit einer [PropertyGroupDescription](#) für unsere *CollectionView* erstellen und schon haben wir Gruppieren.

Aber für welche Eigenschaft? Wir haben keine Eigenschaft zum Gruppieren nach Anfangsbuchstaben. Wir erstellen uns also eine schreibgeschützte Eigenschaft in unserem [ContactViewModel](#).

```

Public ReadOnly Property FirstLastNameLetter As Char
  Get
    Return FullName.First()
  End Get
End Property

```

Da die WPF aber auch für schreibgeschützte Eigenschaften über Änderungen benachrichtigt werden will tun wir das in der Eigenschaft [FullName](#).

```

Public Property FullName() As String
    Get
        Return $"{_modelContact.FirstName} {_modelContact.LastName}"
    End Get
    Set(ByVal value As String)
        Dim fullname = value.Split(CChar(" "))
        _modelContact.FirstName = fullname.First
        If fullname.Length > 1 Then _modelContact.LastName = fullname.Last
        RaisePropertyChanged()
        RaisePropertyChanged(NameOf(FirstLastNameLetter))
    End Set
End Property

```

Gut, jetzt können wir unsere `PropertyGroupDescription` in der `MainWindow.vb` einfügen. Die komplette `MainWindow_Loaded` sieht nun wie folgt aus:

```

Private Sub MainWindow_Loaded(sender As Object, e As RoutedEventArgs) Handles Me.Loaded
    AllContacts = New ObservableCollection(Of ContactViewModel)

    Dim dp As New ContactsDataProvider()
    dp.GetAllContacts(True).ToList.ForEach((Sub(x) AllContacts.Add(New ContactViewModel(x))))


    AllContactsView = CollectionViewSource.DefaultView(AllContacts)
    AllContactsView.Filter = AddressOf Contacts_Filter
    AllContactsView.GroupDescriptions.Add(New PropertyGroupDescription("FirstLastNameLetter"))

    SelectedContact = AllContacts.FirstOrDefault()

    Me.DataContext = Me
End Sub

```

Das Ergebnis kann sich durchaus sehen lassen.



16

Natürlich zeige ich all das auch wieder im Video wobei ich hier noch auf ein paar kleine Details mehr eingehe. Viel Spaß damit.

Hier geht's zum Video: <https://youtu.be/rmpb6fv4LNA>

Hier zum Thread im VB-Paradise-Forum: <https://www.vb-paradise.de/index.php/Thread/124617-Tutorialreihe-WPF-lernen>

Hier zum SupportThread im Forum: <https://www.vb-paradise.de/index.php/Thread/124641-Support-Anregungen-W%C3%BCnsche-zur-Tutorialreihe-WPF-lernen/>

4.4

Einträge bearbeiten und speichern

In diesem Kapitel möchten wir Einträge bearbeiten und speichern. Dies ist ein übliches Szenario in einer jeden CRUD Anwendung. Ich entscheide mich in diesem Fall Daten in einem Dialog zu bearbeiten.

***Info:** Im Video spreche ich davon das wir die Funktion Einträge bearbeiten zu können im nächsten Video noch ausbauen, ich habe mich aber dazu entschieden bereits etwas früher zum Thema MVVM zu springen. Dies wurde von einigen auch so gewünscht.*

Erst benötigen wir in unserem `MainWindow` einen Button welcher dann Dialog dann öffnet. Diesen definiere ich innerhalb der `Grids` und vergebe auch einen Handler „`btnEditContact`“ für das Click-Event:

```
<Button Content="Eintrag bearbeiten" Grid.Column="1" HorizontalAlignment="Right"
        VerticalAlignment="Top" Margin="5" Padding="5"
        Click="btnEditContact"/>
```

```
Private Sub btnEditContact(sender As Object, e As RoutedEventArgs)
End Sub
```

Wir erinnern uns dass wir in der `MainWindow_Loaded` Methode sowohl alle Variablen und Eigenschaften initialisieren als auch die Daten laden. Den Code zum Laden der Daten müssen wir in eine neue Methode auslagern, denn wir möchten ja später sicher nach dem speichern der Daten auch diese wieder neu in die Liste laden, das geht zwar natürlich auch nur mit diesem einen Eintrag, aber der Einfachheit halber laden wir in diesem Beispiel einfach alles neu in die Liste.

***Tipp:** Wie man in Visual Studio sehr einfach über die Refactoring-Funktion Code auslagert seht ihr in dem Video zu diesem Kapitel.*

17

```

Private Sub MainWindow_Loaded(sender As Object, e As RoutedEventArgs) Handles
Me.Loaded
    AllContacts = New ObservableCollection(Of ContactViewModel)

    LoadData()
    Me.DataContext = Me
End Sub

Private Sub LoadData()
    _allContactsModels = dp.GetAllContacts(True).ToList()
    _allContactsModels.ForEach((Sub(x) AllContacts.Add(New ContactViewModel(x))))

    AllContactsView = CollectionViewSource.GetDefaultView(AllContacts)
    AllContactsView.Filter = AddressOf Contacts_Filter
    AllContactsView.GroupDescriptions.Add(New
PropertyGroupDescription("FirstLastNameLetter"))

    SelectedContact = AllContacts.FirstOrDefault()
End Sub

```

Gut, in unserem Handler welcher den „Bearbeiten-Dialog“ öffnen soll müssen wir nun einen Dialog öffnen. Den bauen wir uns schnell mal. Ich baue gerne [UserControls](#) anstatt Fenster da ich damit flexibler bin wenn ich eine Komponente mal austauschen möchte.

```

<UserControl x:Class="uclEditContact"
...
    d:DesignHeight="450" d:DesignWidth="800" Width="400">
<Grid Margin="10">
    <Grid.ColumnDefinitions>
        <ColumnDefinition Width="Auto"/>
        <ColumnDefinition Width="*/>
    </Grid.ColumnDefinitions>
    <Grid.RowDefinitions>
        <RowDefinition Height="50"/>
        <RowDefinition Height="Auto"/>
        <RowDefinition Height="Auto"/>
        <RowDefinition Height="Auto"/>
        <RowDefinition Height="Auto"/>
        <RowDefinition Height="Auto"/>
        <RowDefinition Height="20"/>
        <RowDefinition Height="30"/>
    </Grid.RowDefinitions>
    <Label FontSize="20" Grid.ColumnSpan="2">Eintrag bearbeiten</Label>
    <Label Grid.Row="1">Name</Label>
    <Label Grid.Row="2">Notiz</Label>
    <TextBox Grid.Row="1" Grid.Column="1" Text="{Binding FullName}"/>
    <TextBox Grid.Row="2" Grid.Column="1" Text="{Binding Note}"/>

    <Button Grid.Row="10" Grid.Column="1" Content="Speichern"
Click="btnSaveClick"></Button>
</Grid>
</UserControl>

```

Wie Ihr seht, ist in diesem Dialog nicht viel Spezielles enthalten. Zwei TextBoxen und zwei dazugehörige Labels sowie ein Speicher-Button. Die TextBoxen sind mittels Binding gebunden. Aber an was sich diese gebunden? Woher weiß die WPF dass dieser Dialog einen Telefonbucheintrag bearbeitet?

Dazu kommen wir gleich, erstmal hauchen wir dem Speichern-Button Leben ein indem wir für diesen einen Handler erstellen und folgendes in die Methode packen:

```
Private Sub btnSaveClick(sender As Object, e As RoutedEventArgs)
    Dim win = Window.GetWindow(Me)
    win.DialogResult = True
    win.Close()
End Sub
```

OK, ein wenig habe ich gelogen, es ist ja im Grunde gar kein Button welcher Speichert. Er schließt nur den Dialog, denn es ist einfacher wenn wir von „draußen“ speichern, also im [MainWindow](#), dort haben wir all unsere Daten und können hier leichter speichern. Der Dialog soll und am besten ein DialogResult zurückgeben damit wir außen wissen ob der User auf Speichern geklickt hat oder ob er das Fenster (z.B. mit dem X) geschlossen hat. Das machen wir mittels DialogResult.

Da es sich hier um ein [UserControl](#) handelt und nicht um ein Fenster können wir das Fenster in welchem es sich später befinden wird nicht einfach so aufrufen und dieses schließen.

Mit [Window.GetWindow\(Me\)](#) bekommen wir eine Methode des Frameworks zur Hand welches uns die Instanz des Fensters zurückgibt in welchem sich das Objekt befindet welcher wir unser UserControl als Parameter übergeben.

Wir setzen dann nur den DialogResult und schließen das Fenster.

Aber wie kommt nun die Bindung zustande? Das machen wir im selben Schritt in welchem wir auch das Fenster öffnen lassen. Und zwar in unserer Methode [btnEditContact](#) in unserem MainWindow-CodeBehind.

```
Private Sub btnEditContact(sender As Object, e As RoutedEventArgs)
    Dim dlgWin As New Window()
    dlgWin.SizeToContent = SizeToContent.WidthAndHeight
    dlgWin.Content = New uclEditContact()
    dlgWin.DataContext = SelectedContact
    If dlgWin.ShowDialog() Then
        dp.SaveContacts(_allContactsModels)
    Else
        AllContacts.Clear()
        LoadData()
    End If
End Sub
```

OK, was machen wir hier. Wir Instanzieren hier ein neues [Window](#) und setzen den [Content](#). In diesem Fall unser soeben erstelltes [UserControl](#). Den DataContext des [Window](#) (welchen das UserControl wie wir ja wissen erbt) setzen wir auf den aktuell selektierten Eintrag unserer Auflistung. Durch die Bindung werden die Eingaben ja direkt Synchronisiert und wir müssen einfach nur die ganze Liste speichern falls der [DialogResult = True](#) ist. Falls dem nicht so ist

leeren wir unsere Liste und laden diese neu von der Platte. Wie schon erwähnt ist das nicht der schönste Weg, aber für dieses Tutorial und um die Funktionsweise zu verstehen der einfachste, da wir uns mit komplizierten Lade und Speichermechanismen im Moment nicht aufhalten wollen. Das hält die Beispiele einfach und übersichtlich.

Wenn wir die Anwendung nun starten haben wir die neuen Funktionalitäten zur Verfügung und können Telefonbucheinträge bearbeiten und speichern.

Ich hoffe das war verständlich und wir haben nun genug Vorkenntnisse um zu einem viel spannenderen Thema kommen. Dem MVVM Pattern.

Hier geht's zum Video: <https://youtu.be/Ragc3JIrGHM>

Hier zum Thread im VB-Paradise-Forum: <https://www.vb-paradise.de/index.php/Thread/124617-Tutorialreihe-WPF-lernen>

Hier zum SupportThread im Forum: <https://www.vb-paradise.de/index.php/Thread/124641-Support-Anregungen-W%C3%BCnsche-zur-Tutorialreihe-WPF-lernen/>

4

Das MVVM Pattern

Von vielen schon lange gewünscht kommen wir nun zum richtig spannenden Teil, dem MVVM Muster. Dieses Muster ist ein [Pattern](#) welches die Trennung zwischen Logik und UI ermöglicht wodurch viele Vorteile erlangt werden können, aber wie immer wenn etwas Vorteile hat, hat es auch Nachteile. Auch diese werde ich aufzeigen.

Aber was ist das MVVM Pattern genau? Im Video erkläre ich es genauer und erläutere einige Textpassagen des dazugehörigen [Wikipedia Artikeln zum MVVM Pattern](#).

Kurz und knapp die Vorteile und Nachteile von welchen ich denke das sie für die meisten relevant sein dürften.

Vorteile:

- Die [Geschäftslogik](#) kann unabhängig von der Darstellung bearbeitet werden
- Die Testbarkeit verbessert sich, da die ViewModel die UI-Logiken enthalten und unabhängig von der View instanziiert werden können
- Views können von reinen UI-Designern gestaltet werden, während Entwickler unabhängig davon die Models und ViewModels implementieren

20

Nachteile:

- Für kleinere Projekte eher „overkill“
- Für Anfänger eher nicht geeignet da es schnell komplex wirken kann

Wie Datenbindung an sich und wie diese im MVVM Pattern funktioniert ist auch im verlinkten Wiki Artikel enthalten, ich gehe aber auch auf diesen Part im Video genauer ein, sorry das ich das hier nicht nochmals aufschreibe, ich denke was die Theorie angeht gibt es ja genug Artikel im Netz. Dennoch – im Video gibt’s ein paar Zusatzinfos.

Hier geht’s zum Video: <https://youtu.be/Mw9CJJhpTuA>

Hier zum Thread im VB-Paradise-Forum: <https://www.vb-paradise.de/index.php/Thread/124617-Tutorialreihe-WPF-lernen>

Hier zum SupportThread im Forum: <https://www.vb-paradise.de/index.php/Thread/124641-Support-Anregungen-W%C3%BCnsche-zur-Tutorialreihe-WPF-lernen/>

5.2

Wann MVVM und wann nicht?

Das ist eine Frage die nicht so leicht zu beantworten ist. Zum einen sollte der eigene Wissensstand auf einem gewissen Level sein. Doch auf welchem.

Für MVVM ist einiges notwendig, hierfür sollten Dinge wie Vererbung, Interfaces, generische Klassen und XAML Binding schon recht gut sitzen. Ist dies nicht der Fall hält man sich gut und gerne Wochenlang auf wenn man auf ein Problem stößt. Denn die sind Techniken welche immer wieder zum Einsatz kommen und sollten schon sitzen.

Aber nicht nur das. Immer wieder lese ich in Foren dass man unter WPF unbedingt mit dem MVVM Pattern arbeiten muss. Muss? Why?

Das kann ich so nicht unterschreiben. Es kommt immer auf das Projekt an. Habe ich nur vor ein kleines „Helferlein“ zu proggen und ich weis im Vorfeld das wird nur eine kleine App mit ein bis zwei Funktionen und evtl. zwei bis drei Fenstern, warum soll ich mir dann diesen Overhead antun?

21

Solche Anwendungen sind sicher nicht so komplex das ich von den Vorteilen des Patterns profitieren würde, und die Testbarkeit ist bei solchen Apps auch irrelevant.

Habe ich allerdings vor eine App zu schreiben welche evtl. in Zukunft wächst, wo sich Abhängigkeiten in Zukunft ändern, wo ich im Vorfeld weis das die Anwendung größer wird und die Komplexität immer weiter steigen wird, dann macht MVVM durchaus Sinn da uns das Pattern einfach in gewisser Weise auch zwingt eine gewisse Struktur einzuhalten und somit die Komplexität mit wachsender Anwendung immer weiter fällt.

Das hört sich komisch an aber je größer die Anwendung, desto weniger komplex wird diese durch das Pattern.

Anfangs wirkt eine Anwendung mit MVVM sehr wirr, sehr viele Projekte, sehr viele Klassen, Unmengen an Namespaces und alles befindet sich wo anders. Brainfuck? Zum Teil ja – Anfangs!

Mit zunehmender Anzahl an Logikklassen und immer wachsender Anwendung freut man sich immer mehr über die Trennung und findet alles sofort, denn ohne diese „Ordnung“ hat man irgendwann keinen Überblick mehr. Zugegeben, hält man Ordnung ist alles gut, aber mal ehrlich, wer hält wirklich immer Ordnung, wer weis schon immer genau welcher Code und welche Logik wo genau ist? Bei MVVM stellt sich die Frage oft erst gar nicht.

Wann und wo Ihr MVVM einsetzt entscheidet Ihr. Unsicher? Frag mich einfach, evtl. kann ich dir weiterhelfen.

Hier geht's zum Video: <https://youtu.be/wjKwCn0Myqo>

Hier zum Thread im VB-Paradise-Forum: <https://www.vb-paradise.de/index.php/Thread/124617-Tutorialreihe-WPF-lernen>

Hier zum SupportThread im Forum: <https://www.vb-paradise.de/index.php/Thread/124641-Support-Anregungen-W%C3%BCnsche-zur-Tutorialreihe-WPF-lernen/>